

Lesson 28: Euler-Maclaurin Series and Fourier Series

[> restart;

Sum of a slowly convergent series

The Euler-Maclaurin formula says that an anti-difference $F(x)$ for the function $f(x)$, i.e. a function such that $F(x+1) - F(x) = f(x)$, is given by

$$F(x) = \int f(x) dx + C + \sum_{n=1}^N \frac{\text{bernoulli}(n) D^{(n-1)}(f)(x)}{n!} + R_N(x)$$

where the remainder $R_N(x)$ depends on $D^{(N)}(f)$: if $|D^{(N)}(f)(t)| \leq K(x)$ for $x \leq t \leq x+1$, then $R_N(x) = O(K(x))$ as $x \rightarrow \infty$. The right side can be computed by Maple using the **eulermac** command.

We applied this to approximating the tail of a convergent series: $T(N) = \sum_{n=N}^{\infty} f(n)$ is an antidifference of $-f(N)$. When $f(x)$ and its derivatives go to 0 as $x \rightarrow \infty$, we get the asymptotic formula

$$T(x) = \int_x^{\infty} f(t) dt - \left(\sum_{n=1}^N \frac{\text{bernoulli}(n) D^{(n-1)}(f)(x)}{n!} \right) + O(D^N(f)(x)).$$

I tried this first with $f(x) = \frac{1}{x(x+1)}$:

```
> f1 := x -> 1/(x*(x+1));
```

$$f1 := x \rightarrow \frac{1}{x(x+1)}$$

Typically, successive terms of the Euler-Maclaurin series (after the first few) have opposite signs, and the actual tail $T(x)$ is between the Euler-Maclaurin sums for $2n$ and $2n+2$.

To get the best possible approximation for our sum (with a fixed x) using Euler-Maclaurin series, we take more and more terms until the values stop getting closer together. E.g. for $x=2$:

```
> Digits := 17:
```

```
L := evalf(eval([seq([2*n, eulermac(f1(t), t=x..infinity, 2*n)],
n=1..20)], {x=2, O=0}));
```

```
L := [[2., 0.50037251551557180], [4., 0.49995456284067468], [6.,
0.50001112338784736], [8., 0.49999548241197352], [10., 0.50000275231661133],
[12., 0.49999764239800734], [14., 0.50000271123809833], [16.,
0.49999595792742970], [18., 0.50000759995587279], [20., 0.49998237693343093],
[22., 0.50004947329214925], [24., 0.49983446168064254], [26.,
```

```
0.50065143476794364], [28., 0.49701955469328944], [30., 0.51569498557598687],
[32., 0.40570956769900634], [34., 1.1412047943501407], [36.,
-4.4013161176398831], [38., 42.349796138789312], [40., -396.40358102723478]]
```

```
> [seq([L[j,1],L[j,2]-L[j-1,2]],j=2..20)];
[[4., -0.00041795267489712], [6., 0.00005656054717268], [8.,
-0.00001564097587384], [10., 0.00000726990463781], [12.,
-0.00000510991860399], [14., 0.00000506884009099], [16.,
-0.00000675331066863], [18., 0.00001164202844309], [20.,
-0.00002522302244186], [22., 0.00006709635871832], [24.,
-0.00021501161150671], [26., 0.00081697308730110], [28.,
-0.00363188007465420], [30., 0.01867543088269743], [32.,
-0.10998541787698053], [34., 0.73549522665113436], [36.,
-5.5425209119900238], [38., 46.751112256429195], [40., -438.75337716602409]]

> min(map(t -> abs(t[2]), %));
0.0000050688400909
```

(1.1)

The smallest absolute difference is $j=7$, corresponding to $L[6]$ and $L[7]$

```
> L[6],L[7];
[12., 0.49999764239800734], [14., 0.50000271123809833]
```

That is, $\sum_{j=2}^{\infty} \frac{1}{j(j+1)}$ should be between 0.49999764239800734 (obtained from `eulermac(f1(t),t=x..infinity,12)`) and 0.50000271123809829 (obtained from `eulermac(f1(t),t=x..infinity,14)`).

```
> sum(1/(j*(j+1)),j=2..infinity);
1/2
```

Or if we want to approximate the sum of a series within a given error tolerance ϵ , we could try some particular n and find x such that the difference between `eulermac(f(t), t = x .. infinity, 2*n)` and `eulermac(f(t), t = x .. infinity, 2*n-2)` is less than 2ϵ in absolute value, and use the average of those two `eulermac` values as the approximation for the sum from x to ∞ . Let's try it for our series

$\sum_{k=1}^{\infty} \frac{1}{k^2 + \ln(k)}$ from Lesson 23, with $\epsilon = 10^{-10}$ and $n = 5$, $2n = 10$.

```
> f2 := t -> 1/(t^2 + ln(t));
f2 := t -> 1/(t^2 + ln(t))
```

The difference between the two `eulermac` values will be

```
> Delta := bernoulli(10)*(D@@(9))(f2)(x)/10!;
```

$$\Delta := \frac{5}{264} \frac{\left(2x + \frac{1}{x}\right)^3 \left(2 - \frac{1}{x^2}\right)}{\left(x^2 + \ln(x)\right)^6 x^4} - \frac{1}{176} \frac{\left(2x + \frac{1}{x}\right) \left(2 - \frac{1}{x^2}\right)^2}{\left(x^2 + \ln(x)\right)^5 x^4}$$

$$\begin{aligned}
& + \frac{1}{110} \frac{\left(2x + \frac{1}{x}\right)^2 \left(2 - \frac{1}{x^2}\right)}{(x^2 + \ln(x))^5 x^5} + \frac{1}{264} \frac{\left(2x + \frac{1}{x}\right) \left(2 - \frac{1}{x^2}\right)}{(x^2 + \ln(x))^4 x^6} \\
& + \frac{5}{132} \frac{\left(2x + \frac{1}{x}\right)^4 \left(2 - \frac{1}{x^2}\right)}{(x^2 + \ln(x))^7 x^3} - \frac{5}{264} \frac{\left(2x + \frac{1}{x}\right)^2 \left(2 - \frac{1}{x^2}\right)^2}{(x^2 + \ln(x))^6 x^3} \\
& + \frac{1}{198} \frac{\left(2x + \frac{1}{x}\right) \left(2 - \frac{1}{x^2}\right)}{(x^2 + \ln(x))^5 x^6} - \frac{1}{132} \frac{\left(2x + \frac{1}{x}\right)^9}{(x^2 + \ln(x))^{10}} - \frac{1}{1188} \frac{1}{(x^2 + \ln(x))^2 x^9} \\
& - \frac{19}{11880} \frac{1}{(x^2 + \ln(x))^3 x^9} - \frac{1}{3564} \frac{1}{(x^2 + \ln(x))^4 x^9} \\
& + \frac{1}{33} \frac{\left(2x + \frac{1}{x}\right)^7 \left(2 - \frac{1}{x^2}\right)}{(x^2 + \ln(x))^9} - \frac{7}{396} \frac{\left(2x + \frac{1}{x}\right)^6}{(x^2 + \ln(x))^8 x^3} - \frac{1}{880} \frac{\left(2 - \frac{1}{x^2}\right)^2}{(x^2 + \ln(x))^4 x^5} \\
& - \frac{7}{176} \frac{\left(2x + \frac{1}{x}\right)^5 \left(2 - \frac{1}{x^2}\right)^2}{(x^2 + \ln(x))^8} + \frac{5}{264} \frac{\left(2x + \frac{1}{x}\right)^3 \left(2 - \frac{1}{x^2}\right)^3}{(x^2 + \ln(x))^7} \\
& - \frac{5}{2112} \frac{\left(2x + \frac{1}{x}\right) \left(2 - \frac{1}{x^2}\right)^4}{(x^2 + \ln(x))^6} + \frac{1}{792} \frac{\left(2 - \frac{1}{x^2}\right)^3}{(x^2 + \ln(x))^5 x^3} \\
& - \frac{1}{88} \frac{\left(2x + \frac{1}{x}\right)^5}{(x^2 + \ln(x))^7 x^4} - \frac{1}{132} \frac{\left(2x + \frac{1}{x}\right)^4}{(x^2 + \ln(x))^6 x^5} - \frac{1}{198} \frac{\left(2x + \frac{1}{x}\right)^3}{(x^2 + \ln(x))^5 x^6} \\
& - \frac{1}{308} \frac{\left(2x + \frac{1}{x}\right)^2}{(x^2 + \ln(x))^4 x^7} + \frac{1}{924} \frac{2 - \frac{1}{x^2}}{(x^2 + \ln(x))^3 x^7} - \frac{1}{528} \frac{2x + \frac{1}{x}}{(x^2 + \ln(x))^3 x^8} \\
& - \frac{1}{132} \frac{\left(2x + \frac{1}{x}\right)^2}{(x^2 + \ln(x))^5 x^7} - \frac{47}{10560} \frac{2x + \frac{1}{x}}{(x^2 + \ln(x))^4 x^8} + \frac{1}{528} \frac{2 - \frac{1}{x^2}}{(x^2 + \ln(x))^4 x^7} \\
& - \frac{5}{594} \frac{\left(2x + \frac{1}{x}\right)^3}{(x^2 + \ln(x))^6 x^6}
\end{aligned}$$

Not very pleasant, but what is it approximately?

> `asympt(Delta, x);`

$$O\left(\frac{1}{x^{11}}\right)$$

Can I get that more precisely? It takes a few tries to get the order large enough...

```
> asympt(Delta,x,22);
```

$$-\frac{5}{66x^{11}} + O\left(\frac{1}{x^{13}}\right)$$

OK so to have $\frac{5}{66x^{11}} = 2 \cdot 10^{-10}$ we'd need ...

```
> fsolve(5/(66*x^11)=2*10^(-10),x);
6.0235731458940665
```

6 is not quite good enough, so try $x = 7$.

```
> evalf(eval(Delta,x=7));
-2.89608868472082 10^-11
```

So here are our Euler-Maclaurin values for the tails.

```
> eval(eulermac(f2(t), t = x .. infinity, 10),{x=7, O=0});
```

$$\begin{aligned} & \frac{43839200523987}{3228288560 (49 + \ln(7))^7} - \frac{273970723475691}{40353607 (49 + \ln(7))^9} \\ & + \frac{1130008571789}{958801702320 (49 + \ln(7))^2} - \frac{88309800570123}{230592040 (49 + \ln(7))^8} + \frac{1}{2 (49 + \ln(7))} \\ & + \frac{27682340832837603}{161414428 (49 + \ln(7))^{10}} + \int_7^{\infty} \frac{1}{t^2 + \ln(t)} dt \\ & - \frac{9425751322615127}{402696714974400 (49 + \ln(7))^4} + \frac{15659064687883}{67116119162400 (49 + \ln(7))^3} \\ & + \frac{5499975012585}{2582630848 (49 + \ln(7))^6} - \frac{28259758560269}{639201134880 (49 + \ln(7))^5} \end{aligned}$$

The integral has to be done numerically, but that's no problem for Maple.

```
> E10:= evalf(%);
```

$$E10 := 0.15096255368160152$$

(1.2)

```
> E8:= evalf(eval(eulermac(f2(t), t = x .. infinity, 8),{x=7, O=0}));
```

$$E8 := 0.15096255365264064$$

So our estimate for $\sum_{j=7}^{\infty} \frac{1}{j^2 + \ln(j)}$ is

```
> Tapprox:= evalf((E10+E8)/2);
```

$$Tapprox := 0.15096255366712108$$

and our approximation for $\sum_{j=1}^{\infty} \frac{1}{j^2 + \ln(j)}$ is

```
> evalf(Tapprox + add(f2(j),j=1..6));
```

$$1.5846207083928498$$

Compare to Maple's answer:

```
> evalf(Sum(1/(j^2+ln(j)),j=1..infinity)) - % ;
1.26931 10^-11
```

Consider: we started by saying we'd need 10 billion terms of the series to have a partial sum accurate to within 10^{-10} . Then by estimating the tail using integrals we used 70711 terms. The improvement on this using the Midpoint Rule used 224 terms. And now with Euler-Maclaurin we just used 6 terms. Impressive, no?

A slowly divergent series

Our second application of Euler-Maclaurin will be to a slowly divergent series:

```
> S := N -> sum(1/n, n=1..N);
```

$$S := N \rightarrow \sum_{n=1}^N \frac{1}{n} \quad (2.1)$$

What does Euler-Maclaurin say this partial sum is asymptotic to?

```
> R := eulermac(1/n, n=1..N, 10);
```

$$R := \int_1^N \frac{1}{n} dn + \gamma + \frac{1}{2N} - \frac{1}{12N^2} + \frac{1}{120N^4} - \frac{1}{252N^6} + \frac{1}{240N^8} - \frac{1}{132N^{10}} + O\left(\frac{1}{N^{12}}\right) \quad (2.2)$$

```
> S1 := unapply(eval(%), 0=0), N) assuming N >= 1;
```

$$S1 := N \rightarrow \ln(N) + \gamma + \frac{1}{2N} - \frac{1}{12N^2} + \frac{1}{120N^4} - \frac{1}{252N^6} + \frac{1}{240N^8} - \frac{1}{132N^{10}} \quad (2.3)$$

Check it out for $N=5$.

```
> evalf([S1(5), S(5)]);
```

$$[2.2833333332582884, 2.2833333333333333] \quad (2.4)$$

```
> %[1]-%[2];
```

$$-7.50449 10^{-11} \quad (2.5)$$

There's a little bit of "magic" here. Euler-Maclaurin is supposed to have an arbitrary constant. How does Maple choose that the constant term γ ?

```
> ?gamma
```

For less famous series, **eulermac** might not give us the constant. Let's try:

```
> S := N -> sum(j/(j^2+1), j=1..N);
```

$$S := N \rightarrow \sum_{j=1}^N \frac{j}{j^2+1} \quad (2.6)$$

```
> R8 := eulermac(j/(j^2+1), j=1..N, 8);
```

$$R8 := \frac{1}{2} \ln(N^2+1) - \frac{5953}{55440} + O(1) + \frac{1}{2} \frac{N}{N^2+1} + \frac{1}{12(N^2+1)} - \frac{1}{6} \frac{N^2}{(N^2+1)^2} \quad (2.7)$$

$$\begin{aligned}
& -\frac{1}{15} \frac{N^2}{(N^2+1)^3} + \frac{1}{120} \frac{1}{(N^2+1)^2} + \frac{1}{15} \frac{N^4}{(N^2+1)^4} + \frac{4}{21} \frac{N^4}{(N^2+1)^5} \\
& -\frac{1}{14} \frac{N^2}{(N^2+1)^4} + \frac{1}{252} \frac{1}{(N^2+1)^3} - \frac{8}{63} \frac{N^6}{(N^2+1)^6} - \frac{16}{15} \frac{N^6}{(N^2+1)^7} \\
& + \frac{2}{3} \frac{N^4}{(N^2+1)^6} - \frac{2}{15} \frac{N^2}{(N^2+1)^5} + \frac{1}{240} \frac{1}{(N^2+1)^4} + \frac{8}{15} \frac{N^8}{(N^2+1)^8} + O\left(\right. \\
& \left. -\frac{1280 N^8}{(N^2+1)^9} + \frac{1120 N^6}{(N^2+1)^8} - \frac{400 N^4}{(N^2+1)^7} + \frac{50 N^2}{(N^2+1)^6} - \frac{1}{(N^2+1)^5} \right. \\
& \left. + \frac{512 N^{10}}{(N^2+1)^{10}} \right)
\end{aligned}$$

This one has O(1) to indicate the constant term.

> eval(R8, {O=0});

$$\begin{aligned}
& \frac{1}{2} \ln(N^2+1) - \frac{5953}{55440} + \frac{1}{2} \frac{N}{N^2+1} + \frac{1}{12} \frac{1}{(N^2+1)} - \frac{1}{6} \frac{N^2}{(N^2+1)^2} \\
& -\frac{1}{15} \frac{N^2}{(N^2+1)^3} + \frac{1}{120} \frac{1}{(N^2+1)^2} + \frac{1}{15} \frac{N^4}{(N^2+1)^4} + \frac{4}{21} \frac{N^4}{(N^2+1)^5} \\
& -\frac{1}{14} \frac{N^2}{(N^2+1)^4} + \frac{1}{252} \frac{1}{(N^2+1)^3} - \frac{8}{63} \frac{N^6}{(N^2+1)^6} - \frac{16}{15} \frac{N^6}{(N^2+1)^7} \\
& + \frac{2}{3} \frac{N^4}{(N^2+1)^6} - \frac{2}{15} \frac{N^2}{(N^2+1)^5} + \frac{1}{240} \frac{1}{(N^2+1)^4} + \frac{8}{15} \frac{N^8}{(N^2+1)^8}
\end{aligned} \tag{2.8}$$

> S8:= unapply(%, N);

$$\begin{aligned}
S8 := N \rightarrow & \frac{1}{2} \ln(N^2+1) - \frac{5953}{55440} + \frac{1}{2} \frac{N}{N^2+1} + \frac{1}{12} \frac{1}{(N^2+1)} - \frac{1}{6} \frac{N^2}{(N^2+1)^2} \\
& -\frac{1}{15} \frac{N^2}{(N^2+1)^3} + \frac{1}{120} \frac{1}{(N^2+1)^2} + \frac{1}{15} \frac{N^4}{(N^2+1)^4} + \frac{4}{21} \frac{N^4}{(N^2+1)^5} \\
& -\frac{1}{14} \frac{N^2}{(N^2+1)^4} + \frac{1}{252} \frac{1}{(N^2+1)^3} - \frac{8}{63} \frac{N^6}{(N^2+1)^6} - \frac{16}{15} \frac{N^6}{(N^2+1)^7} \\
& + \frac{2}{3} \frac{N^4}{(N^2+1)^6} - \frac{2}{15} \frac{N^2}{(N^2+1)^5} + \frac{1}{240} \frac{1}{(N^2+1)^4} + \frac{8}{15} \frac{N^8}{(N^2+1)^8}
\end{aligned} \tag{2.9}$$

So S8 should have the property that S8(N) - S(N) goes to a constant as N -> infinity, but we don't know the value of the constant. The next term of the Euler-Maclaurin series would be

> eval(eulermac(j/(j^2+1), j=1..N, 10) - eulermac(j/(j^2+1), j=1..N, 8), O=0);

$$-\frac{691}{32760} + \frac{1}{132} \frac{1}{(N^2+1)^5} + \frac{320}{33} \frac{N^8}{(N^2+1)^9} - \frac{280}{33} \frac{N^6}{(N^2+1)^8} \tag{2.10}$$

$$+ \frac{100}{33} \frac{N^4}{(N^2+1)^7} - \frac{25}{66} \frac{N^2}{(N^2+1)^6} - \frac{128}{33} \frac{N^{10}}{(N^2+1)^{10}}$$

I must admit I'm puzzled by that constant $-\frac{691}{32760}$. It's apparently using a different arbitrary constant. Anyway, if we take that out:

```
> D8:= % + 691/32760;
```

$$D8 := \frac{1}{132} \frac{1}{(N^2+1)^5} + \frac{320}{33} \frac{N^8}{(N^2+1)^9} - \frac{280}{33} \frac{N^6}{(N^2+1)^8} + \frac{100}{33} \frac{N^4}{(N^2+1)^7} - \frac{25}{66} \frac{N^2}{(N^2+1)^6} - \frac{128}{33} \frac{N^{10}}{(N^2+1)^{10}} \quad (2.11)$$

```
> D8-bernoulli(10)*diff(N/(N^2+1), N$9)/10!;
```

$$0 \quad (2.12)$$

```
> evalf(eval(D8,N=10));
```

$$-3.9146147587712791 \cdot 10^{-13} \quad (2.13)$$

So if we compare S8 and the actual partial sum S at N = 10, we should get a good value for the constant.

```
> C:= evalf(S(10)-S8(10));
```

$$C := 0.0127270242544835 \quad (2.14)$$

Thus a good approximation for S(N) when N is large should be C + S8(N).

For example, what's the first N for which S(N) > 13?

```
> fsolve(C+S8(N)=13,N=1..infinity);
```

$$4.8633321354098285 \cdot 10^5 \quad (2.15)$$

Apparently it should be 486334.

```
> evalf(S(486334));
```

$$13.000001617117108 + 0. I \quad (2.16)$$

```
> evalf(S(486333));
```

$$12.999999560917048 + 0. I \quad (2.17)$$

These imaginary parts (even though 0) are rather suspicious. It turns out that Maple has a formula for the sum, involving complex quantities.

```
> S(N);
```

$$\frac{1}{2} \Psi(N+1-I) + \frac{1}{2} \Psi(N+1+I) - \frac{1}{2} \Psi(1-I) - \frac{1}{2} \Psi(1+I) \quad (2.18)$$

We could try actually adding the numbers (using evalhf to make it faster).

```
> evalhf(add(j/(j^2+1),j=1..486333));
```

$$12.9999995609168568 \quad (2.19)$$

```
> evalhf(add(j/(j^2+1),j=1..486334));
```

$$13.0000016171169168 \quad (2.20)$$

Fourier Series

A Fourier series is a quite different type of series. Instead of a series in powers of the variable x , it is a series in trigonometric functions of x .

Let f be a periodic function of one variable, i.e. there is some T (the period) such that $f(x + T) = f(x)$ for all x . For convenience, I'll suppose the period is 2π . Examples of functions with a period of 2π are $\sin(nx)$ and $\cos(nx)$ for integers n . Those are the basic building blocks of the Fourier series. We sometimes consider functions f that are initially defined on just one interval of length 2π , say $[0, 2\pi)$, and then extend them to all other real numbers using the periodicity. For example,

```
> f := t -> t^2;
```


$$f := t \rightarrow t^2 \quad (3.1)$$

This is not periodic. We want to make it periodic. So e.g. for $2\pi \leq x < 4\pi$ we should take $f(x) = f(x - 2\pi)$, and for $-4\pi \leq x < -2\pi$ we should take $f(x) = f(x + 4\pi)$. The following "sawtooth" function helps. It takes any x to the number in the interval $[0, 2\pi)$ that differs from it by a multiple of 2π .

```
> saw := x -> x - 2*Pi*floor(x/(2*Pi));
```

$$saw := x \rightarrow x - 2\pi \operatorname{floor}\left(\frac{1}{2} \frac{x}{\pi}\right) \quad (3.2)$$

```
> plot(saw(x), x=-10..10, discontinuous);
```


The periodic version of f is then $f(\text{saw}(x))$.

```
> fper := f @ saw;
```

```
fper := f@saw
```

(3.3)

```
> plot(fper(x), x=-10 .. 10, discontin=true);
```



The Fourier series of f is $\frac{a_0}{2} + \sum_{k=1}^{\infty} (a_k \cos(kx) + b_k \sin(kx))$ where

$$a_k = \frac{1}{\pi} \int_0^{2\pi} f(x) \cos(kx) dx \text{ and } b_k = \frac{1}{\pi} \int_0^{2\pi} f(x) \sin(kx) dx.$$

The theoretical result is that for a "nice" function f this Fourier series converges to the periodic version $f_{per}(x)$ for every x where f_{per} is continuous, while at points where f_{per} has a jump discontinuity with limits from the left and right, it converges to the average of those limits. In this case, "nice" means piecewise continuous and piecewise continuously differentiable.

Let's try it with our f . Maple should be able to do these integrals.

```
> a := unapply( 1/Pi * int(f(t)*cos(k*t),t=0..2*Pi),k);
```

$$a := k \rightarrow \frac{4(-\sin(\pi k) \cos(\pi k) + 2k^2 \pi^2 \sin(\pi k) \cos(\pi k) + 2\pi k \cos(\pi k)^2 - \pi k)}{\pi k^3} \quad (3.4)$$

Oops, better tell Maple that k is an integer.

```
> a := unapply( 1/Pi * int(f(t)*cos(k*t),t=0..2*Pi),k) assuming
k::integer;
```

(3.5)

$$a := k \rightarrow \frac{4}{k^2} \quad (3.5)$$

Hmm. What about $k = 0$?

```
> a(0);
```

Error, (in a) numeric exception: division by zero

This points out a common difficulty with Maple: the "specialization problem". Maple will often produce results that are valid for "generic" values of some parameter, but don't work for some specific values. Thus Maple will happily divide something by a variable, without worrying that you might want to take that variable to be 0. It might be very annoying if Maple did always worry about that, and maybe ask you whether the variable could be 0. But you do sometimes encounter those special cases, and this is an example. You have to treat them specially.

```
> a(0) := 1/Pi * int(f(t)*cos(0*t), t=0..2*Pi);
```

$$a(0) := \frac{8}{3} \pi^2 \quad (3.6)$$

This actually defines an entry in what's called a "remember table" for the function a . If you now ask for $a(0)$, instead of using the normal definition Maple will remember the value in this table, and use it.

```
> a(k), a(3), a(0);
```

$$\frac{4}{k^2}, \frac{4}{9}, \frac{8}{3} \pi^2 \quad (3.7)$$

```
> b := unapply( 1/Pi * int(f(t)*sin(k*t), t=0..2*Pi), k) assuming
k::integer;
```

$$b := k \rightarrow -\frac{4\pi}{k} \quad (3.8)$$

No problem here, since we don't use $b(0)$.

```
> Fseries := a(0)/2 + Sum(a(k)*cos(k*x)+b(k)*sin(k*x), k=1..
infinity);
```

$$Fseries := \frac{4}{3} \pi^2 + \sum_{k=1}^{\infty} \left(\frac{4 \cos(kx)}{k^2} - \frac{4\pi \sin(kx)}{k} \right) \quad (3.9)$$

Maple can actually do this sum in "closed form", though it doesn't look much like $f(x)$.

```
> value(Fseries);
```

$$\frac{4}{3} \pi^2 + 2 \operatorname{polylog}(2, e^{ix}) + 2 \operatorname{polylog}(2, e^{-ix}) + 2 I \pi (-\ln(1 - e^{ix}) + \ln(1 - e^{-ix})) \quad (3.10)$$

```
> simplify(% - x^2) assuming x>=0, x<2*Pi;
```

$$\frac{4}{3} \pi^2 + 2 \operatorname{polylog}(2, e^{ix}) + 2 \operatorname{polylog}(2, e^{-ix}) - 2 I \pi \ln(1 - e^{ix}) + 2 I \pi \ln(1 - e^{-ix}) - x^2 \quad (3.11)$$

But we're more interested in the partial sums.

Maple objects introduced in this lesson:

gamma
option remember

